

Lieu de points (1)

Soit A, B, C, D un rectangle. Quel est le lieu des points M tels que les rayons des cercles circonscrits à ABM et CDM soient égaux ?

Commençons par définir les quatre sommets du rectangle dans un repère orthonormé bien choisi.

```
> (A:[a,b],B:[-a,b],C:[-a,-b],D:[a,-b],[ 'A=A, 'B=B, 'C=C, 'D=D ] );
```

$$[A = [a, b], B = [-a, b], C = [-a, -b], D = [a, -b]]$$

Maintenant un point M générique :

```
> M:[x,y];
```

$$[x, y]$$

Pour déterminer le centre du cercle \mathcal{C}_1 circonscrit au triangle ABM , nous allons procéder ainsi :

1/ Définir une fonction f , caractéristique de \mathcal{C}_1 au sens où si $M(x, y) \in \mathcal{C}_1$ alors $f(x, y) = 0$ est l'équation cartésienne de \mathcal{C}_1 .

```
> f(p):=buildq([X:part(p,1),Y:part(p,2)],X^2+Y^2+alpha*X+beta*Y+vgamma)$
```

Cette fonction dépend de trois paramètres à partir desquels se calculent le centre et le rayon du cercle.

2/ Déterminer les paramètres pour que la fonction soit caractéristique du cercle passant par A, B, M . Cela revient à résoudre un système de trois équations linéaires à trois inconnues.

```
> s1:solve([f(A),f(B),f(M)],[alpha,beta,vgamma]);
```

$$\left[\left[\alpha = 0, \beta = -\frac{-y^2 - x^2 + b^2 + a^2}{b - y}, \text{vgamma} = \frac{b(-y^2 - x^2) + b^2 y + a^2 y}{b - y} \right] \right]$$

Nous trouvons $\alpha = 0$, dans le cas qui nous intéresse ce résultat était attendu, le centre du cercle étant nécessairement situé sur l'axe des ordonnées.

Récupérons l'ordonnée du centre.

```
> y1:-1/2*rhs(part(s1,1,2));
```

$$\frac{-y^2 - x^2 + b^2 + a^2}{2(b - y)}$$

Nous pouvons maintenant calculer le carré du rayon du cercle \mathcal{C}_1 .

```
> r1c:(x-0)^2+(y-y1)^2;
```

$$\left(y - \frac{-y^2 - x^2 + b^2 + a^2}{2(b - y)} \right)^2 + x^2$$

Pour le cercle \mathcal{C}_2 passant par C , D et M , il est inutile de reprendre tous les calculs, il suffit de substituer $-b$ à b .

```
> r2c:subst(-b,b,r1c);
```

$$\left(y - \frac{-y^2 - x^2 + b^2 + a^2}{2(-y - b)} \right)^2 + x^2$$

Les points M recherchés doivent vérifier l'égalité des quantités ci-dessus, autrement exprimée à l'aide de MAXIMA de la façon suivante :

```
> (eq:factor(num(radcan(r1c-r2c))),eq=0);
```

$$-by(y^2 - x^2 - b^2 + a^2)(y^2 + x^2 - b^2 - a^2) = 0$$

Nous trouvons deux sous-ensembles *attendus* : la droite d'équation $y = 0$ et le cercle d'équation $x^2 + y^2 = a^2 + b^2$ circonscrit à $ABCD$. Il y en a un d'*imprévu* qui est une hyperbole équilatère, éventuellement dégénérée en deux droites si le rectangle est carré!

Récupérons l'expression définissant l'hyperbole.

```
> h:part(eq,1,3);
```

$$y^2 - x^2 - b^2 + a^2$$

Traisons le cas particulier $a = 3$ et $b = 2$.


```
> load("reduction.mc")$
```

```
> Reduction(ev(h,a=3,b=2));
```

C'est une hyperbole !

$$\frac{X^2}{5} - \frac{Y^2}{5} = 1$$

> Representation(-5,-5,5,5,1);



Il reste maintenant à établir que si M appartient à l'un de ces trois ensembles, la droite, le cercle ou l'hyperbole alors il vérifie la propriété de l'énoncé. On exclura les cas particuliers où M coïncide avec l'un des quatre points A, B, C ou D ; il y a indétermination sur l'un des cercles circonscrits.

Dans le cas des deux premiers ensembles, la propriété est évidente. Pour l'hyperbole, il suffit de considérer que l'on a :

$$x^2 + b^2 = a^2 + y^2$$

Ainsi si $C_1(0, y + b)$ et $C_2(0, y - b)$ alors :

$$C_1M = C_1A = C_1B = C_2M = C_2C = C_2D = \sqrt{x^2 + b^2} = \sqrt{a^2 + y^2}$$

C_1 et C_2 sont les centres des cercles circonscrits à ABM et CDM , ils ont le même rayon !

On remarquera que dans ce dernier cas \mathcal{C}_1 est le *translaté* de \mathcal{C}_2 selon le vecteur $(0, 2b)$.